Arduino Hand's-On

CS5968 / FA3800

Disclaimer

- Many of these slides are mine
- But, some are stolen from various places on the web
  - todbot.com – Bionic Arduino and Spooky Arduino class notes from Tod E.Kurt
  - ladyada.net – Arduino tutorials by Limor Fried

Part 1 – Arduino SW



Remember, Arduino calls programs "sketches"

Part 1 – Arduino SW

Procedure

Using Arduino

- Write your sketch
- Press Compile button (to check for errors)
- Press Upload button to program Arduino board with your sketch

Try it out with the "Blink" sketch!

Load "File/Sketchbook/Examples/Digital/Blink"



Get the Blink Example



Blink Sketch (program)

```
/*
* Blink
*
* The basic Arduino example.  Turns on an LED on for one second,
* then off for one second, and so on...  We use pin 13 because,
* depending on your Arduino board, it has either a built-in LED
* or a built-in resistor so that you need only an LED.
*/

int ledPin = 13;                  // LED connected to digital pin 13

void setup() {                    // run once, when the sketch starts
    pinMode(ledPin, OUTPUT);      // sets the digital pin as output
}

void loop()                       // run over and over again
{
   digitalWrite(ledPin, HIGH);    // sets the LED on
   delay(1000);                   // wait for a second
   digitalWrite(ledPin, LOW);     // sets the LED off
   delay(1000);                   // wait for a second
}
```

Variables

```
int ledPin = 13;      // LED connected to digital pin 13
```

- ledPin is a variable that holds a 16-bit value
  - 16 binary digits is enough for -32768 to 32767
  - Default starting value is 13
  - There are other data types you can use

Data Types on Arduino

- By default, types are signed unless you say "unsigned"…

Type	Size (bits)	Size (bytes)	Minimum	Maximum
boolean	1	1	0 (false)	1 (true)
unsigned byte	8	1	0	255
byte	8	1	-128	127
unsigned int	16	2	0	65,535
int	16	2	-32,768	32,767
unsigned long	32	4	0	4,294,967,295
long	32	4	-2,147,483,648	-2,147,483,647
float (double)	32	4	-3.4028235E+38	3.4028235E+38

Functions

```
void setup() {                  // run once, when the sketch starts
    pinMode(ledPin, OUTPUT); // sets the digital pin as output
    }

<return-type> <function-name> ( <arguments>) {
    <function-body>
    }
```

- "void" means no value is returned
- pinMode(ledPin, OUTPUT); // call another function

Pseudo-code Silly Function

```
clean-cat wash-the-cat ( dirty-cat cat)     // a procedure for washing the cat
    {
    turn on the shower.
    find the cat.
    grab the cat.
    put cat under shower.
    wait 3 minutes.                         // wait for cat to get clean.
    release clean-cat.
    }
```

Required Arduino Functions

```
void setup() {                              // run once, when the sketch starts
    <initialization statement>;             // typically pin definitions
    …                                       // and other init stuff
    <initialization statement>;
}

void loop() {                               // run over and over again
   <main loop statement>;                   // the guts of your program
    …                                       // which could include calls
   <main loop statement>;                   // to other functions…
}
```

Arduino Language (C/C++)

- pinMode(pin,mode);
  - pin is a number, mode can be INPUT or OUTPUT
- digitalWrite(pin, value);
  - Value can be HIGH (1) or LOW (0)
- digitalRead(pin);
  - Returns an int – value either HIGH or LOW
- delay(val);
  - Pauses for val milliseconds (1/1000's of a sec)
  - val can be up to unsigned long max
- millis();
  - Returns number of milliseconds since the program started running
  - Returns an unsigned long – overflows in ~50 days

Blink Sketch (program)

```
/*
* Blink
*
* The basic Arduino example.  Turns on an LED on for one second,
* then off for one second, and so on...  We use pin 13 because,
* depending on your Arduino board, it has either a built-in LED
* or a built-in resistor so that you need only an LED.
*/

int ledPin = 13;                // LED connected to digital pin 13

void setup() {                  // run once, when the sketch starts
    pinMode(ledPin, OUTPUT);    // sets the digital pin as output
}

void loop()                     // run over and over again
{
   digitalWrite(ledPin, HIGH);  // sets the LED on
   delay(1000);                 // wait for a second
   digitalWrite(ledPin, LOW);   // sets the LED off
   delay(1000);                 // wait for a second
}
```

Blink Modifiations

- Change so that blink is on for 50msec and off for 50msec
  - What happens?
- Change so that blink is on for 10ms and off for 10ms
  - What happens?
- Change to use an external LED rather than the one on the board
  - Pay attention to current! Use a current-limiting resistor!

Making Circuits





heart pumps, blood flows

voltage pushes, current flows

Wiring it Up







wiring diagram schematic wiring it up

Electricity flows in a loop. Can stop flow by breaking the loop

Wiring it Up



wiring diagram



schematic

Arduino Diecimila board has this circuit built-in
To turn on LED use `digitalWrite(13,HIGH)`

External LED

- Remember Ohm's Law
  - V = IR I = V/R R = V/I



- Every LED has a Vf "Forward Voltage"
  - How much voltage is dropped passing through the LED

- R = (V – Vf) / I
  - Example – If Vf is 1.9v (red LED), and V = 5v, and you want 15mA of current (0.015A)
  - R = (5 – 1.9)/0.015 = 3.1/0.015 = 206Ω
  - Exact isn't critical – use next size up, i.e. 220Ω
  - Or be safe and use 330Ω or 470Ω
  - This would result in 9.4mA or 6.6mA which is fine

LEDs and Resistors





Current flows from Anode to Cathode
Lights up when current flows

Proto Boards



AKA Solderless Breadboards

Wire it Up



Wire it Up





Resistor Color Codes



What's the color code for a 220Ω resistor?

What's the color code for a 1kΩ resistor?

What's the color code for a 470Ω resistor

Resistor Color Codes



We're using 4-band 5% resistors with a ¼ watt rating

What's the color code for a 220Ω resistor?

What's the color code for a 1kΩ resistor?

What's the color code for a 470Ω resistor

Resistor Color Codes



We're using 4-band 5% resistors with a ¼ watt rating

What's the color code for a 220Ω resistor?

red red brown gold

What's the color code for a 1kΩ resistor?

brown black red gold

What's the color code for a 470Ω resistor

yellow violet brown gold

Wire it Up

- Wire up an external LED of your choice, and change the Blink program to use that external LED
  - Choose your resistor based on the Vf of the LED you're using



Blink Subtlety

- When the delay(val); function runs, nothing else can happen
  - Arduino just sits there counting milliseconds
  - For blink this is just fine, but later you may want other things to be going on while the Arduino is counting
  - Load BlinkWithoutDelay from the examples
  - Let's look at what it does…
  - C "if" statement
    - if (condition) { do something};
    - if (condition) {do something}
      else {do something else};

BlinkWithoutDelay

```
const int ledPin = 13;          // const says this won't change
int ledState = LOW;             // used to set the state of the LED
long previousMillis = 0;        // used to store last time LED changed
long interval = 1000;           //interval at which to blink the LED

void setup() {
      pinMode(ledPin, OUTPUT);                      // set LED pin mode
}

void loop () {
      // check to see if it's time to change the LED value
      if (millis() – previousMillis > interval) {
            previousMillis = millis();              // save the time you made the change
            if (ledState == LOW) { ledState = HIGH; }  // toggle the state of the LED
            else { ledState = LOW; } ;
            digitalWrite(ledPin, ledState);         // set the LED with ledState
      }

      // you can do other things here if it's not time to change the LED state
}
```

Comparison Operators

x == y (x is equal to y)

x != y (x is not equal to y)

x < y (x is less than y)

x > y (x is greater than y)

x <= y (x is less than or equal to y)

x >= y (x is greater than or equal to y)

Beware of x=y; This does an assignment, not a comparison!

Moving on…

Varying LED Brightness

Same circuit as Blink circuit but pin 9 instead of pin 13



schematic wiring diagram wired up

The PWM pins work with the "`analogWrite(value)`" command
where "value" ranges between 0 and 255.
To turn LED to half-bright, use `analogWrite(9,128)`

Pulse Width Modulation

- analogWrite(pin, value);
  - value can be 0 to 255
  - Must be one of the "PWM pins" : pins 3, 5, 6, 9, 10, 11
  - Don't need to set pinMode to OUTPUT (but won't hurt)

Load "File/Sketchbook/Examples/Analog/Fading"

note →

```
int value = 0;                          // variable to keep the actu
int ledpin = 9;                         // light connected to digito

void setup()
{
  // nothing for setup
}

void loop()
{
  for(value = 0 ; value <= 255; value+=5) // fade in (from min to max
  {
    analogWrite(ledpin, value);           // sets the value (range fro
    delay(30);                            // waits for 30 milli second
  }
  for(value = 255; value >=0; value-=5)   // fade out (from max to min
  {
    analogWrite(ledpin, value);
    delay(30);
  }
```

C "for" loop

```
for (initialization; condition; increment) {
    // do something

}


int i;    // define an int to use as a loop variable

for (i = 0; i <= 255; i++) {
    analogWrite(pin, i);
    delay(50);

}
```

C Compound Operators

```
x = x + 1;

x += 5;          // same as x = x + 5

x++;             // same as x = x + 1

x = x – 2;

x -= 3;          // same as x = x - 3

x--;             // same as x = x - 1

x = x * 3;

x *=5;           // same as x = x * 5
```

Moving on…

- Write a program to make the LED flicker like a flame
  - Choose a random intensity
  - For a random amount of time
  - Use analogWrite(ledPin, val)
  - Also, random(min,max); will return a random number between min and max.
    - randomSeed(int); will initialize the random function

Candle Program

```
int ledPin = 9;                 // select pin for LED output
int bright = 0;                 // Variable to hold LED brightness
int time = 0;                   // variable to hold delay time

void setup () {
      randomSeed(0);                    // initialize the random function
      pinMode(ledPin, OUTPUT);          // LED should be output
}

void loop() {
      bright = random(100, 255);        // random brightness value
      analogWrite(ledPin, bright);      // set the LED brightness

      time = random(10,150);            // random time in ms
      delay(time);                      // delay for that time
}
```

Silly LED Tricks



LED Wiring – 2 ways



Getting Input (Digital)

- Switches make or break a connection
- But Arduino wants to see a voltage
  - Specifically, a "HIGH" (5 volts)
  - or a "LOW" (0 volts)

How do you go from make/break to high/low?

Switches

- Digital inputs can "float" between 0 and 5 volts
- Resistor "pulls down" input to ground (0 volts)
- Pressing switch sets input to 5 volts
- Press is HIGH
  Release is LOW



"pull-down"

Why do we need the "pull down" resistor?

Another Switch

- Resistor pulls up input to 5 volts
- Switch sets input to 0 volts
- But now the sense is inverted
  - Press is LOW
  - Release is HIGH



"pull-up"

A Switch



Pressing the button, "closes the gap"

Using a Switch





Using digitalRead()

- In `setup()`: use `pinMode(myPin,INPUT)` to make pin an input
- In `loop()`: use `digitalRead(myPin)` to get switch position
  - If doing many tests, use a variable to hold the output value of `digitalRead()`.
  - `e.g. val = digitalRead(myPin)`

digital_read

Load "examples/digital IO/digital_read"

```
int ledPin = 13; // choose the pin for the LED
int inPin = 7;   // choose the input pin (for a pushbutton)
int val = 0;     // variable for reading the pin status

void setup() {
  pinMode(ledPin, OUTPUT);  // declare LED as output
  pinMode(inPin, INPUT);    // declare pushbutton as input
}

void loop(){
  val = digitalRead(inPin);  // read input value
  if (val == HIGH) {         // check if the input is HIGH (button released)
    digitalWrite(ledPin, LOW);  // turn LED OFF
  } else {
    digitalWrite(ledPin, HIGH);  // turn LED ON
  }
}
```

Now you control the blinking

Moving on…

- Write a program that reads the value on an input pin
  - Use the button to change from blinking fast to blinking slow

Moving on…

- Write a program that reads the value on an input pin
  - Use the button to change from blinking fast to blinking slow

```
int ledPin = 13; // choose the pin for the LED
int inPin = 7;   // choose the input pin (for a pushbutton)
int val = 0;     // variable for reading the pin status
int delayval = 100;

void setup() {
  pinMode(ledPin, OUTPUT);  // declare LED as output
  pinMode(inPin, INPUT);    // declare pushbutton as input
}

void loop(){
  val = digitalRead(inPin);  // read input value

  if( val == HIGH )
    delayval = 1000;
  else
    delayval = 100;

  digitalWrite(ledPin, HIGH);  // blink the LED and go OFF
  delay(delayval);
  digitalWrite(ledPin, LOW);
  delay(delayval);
}
```

Multiple Switches

Same sub-circuit, just duplicate





Make Your Own Switches

- Anything that makes a connection
- Wires, tin foil, tinfoil balls, ball bearings
- Pennies!
- Nails, bolts, screws

- Or repurpose these tiny switches as bump detectors or closure detectors

Make Your Own Switches



Analog Input

To computers, analog is chunky



Analog Input

- Many states, not just two (HIGH/LOW)
- Number of states (or "bins") is *resolution*
- Common computer resolutions:
  - 8-bit = 256 states
  - 16-bit = 65,536 states
  - 32-bit = 4,294,967,296 states



Analog Input on Arduino

- Our version uses ATMega328p
  - six ADC inputs (Analog to Digital Converter)
  - Voltage range is 0-5v
  - Resolution is 10 bits (digital values between 0-1023)
  - In other words, 5/1024 – 4.8mV is the smallest voltage change you can measure
- analogRead(pin);
  - reads an analog pin
  - returns a digital value between 0-1023
  - analog pins need no pinMode declaration



Analog Input

Sure sure, but how to make a varying voltage?

With a *potentiometer*. Or just *pot*.



Potentiometers

Moving the knob is like moving
where the arrow taps the voltage on the resistor



Arduino Analog Input

Red to Vcc

Purple to A0

Blue to Gnd



Analog Input Sketch

Sketch "Examples/sensors_resistive/analog_read_led"

Change to 0

```
int potPin = 2;    // select the input pin for the potentiometer
int ledPin = 13;   // select the pin for the LED
int val = 0;       // variable to store the value coming from the sensor

void setup() {
  pinMode(ledPin, OUTPUT);  // declare the ledPin as an OUTPUT
}

void loop() {
  val = analogRead(potPin);    // read the value from the sensor
  digitalWrite(ledPin, HIGH);  // turn the ledPin on
  delay(val);                  // stop the program for some time
  digitalWrite(ledPin, LOW);   // turn the ledPin off
  delay(val);                  // stop the program for some time
}
```

Turn knob to vary blink rate of the LED
Notice no `pinMode()` for analog inputs

Moving on…

- Write a program to read an analog value from a pot and use that value to control the brightness of an LED
  - Fade the LED by turning the pot
  - Useful function is
    map(value, fromlow, fromhigh, tolow, tohigh);

    y = map(x, 0, 1023, 50, 150);
  - Also remember
    analogWrite(pin,value);
    - PWM value from 0-255



potFade

```
int potPin = 0;          // the analog input pin from the pot
int ledPin = 9;          // pin for LED (a PWM pin)
int val;                 // Variable to hold pot value

void setup () {
    pinMode(ledPin, OUTPUT);          // declare ledPin as output
}

void loop() {
    val = analogRead(potPin);         //read the value from the pot
    val = map(val, 0, 1023, 100, 255);  // map to reasonable values
    analogWrite(ledPin, val);
}
```

What good are pots?

- Anytime you need a ranged input
  - (we're used to knobs)
- Measure rotational position
  - steering wheel, etc.
- But more importantly for us, potentiometers are a good example of a *resistive sensor*

Sensing the Dark

- Pots are example of a *voltage divider*
- Voltage divider splits a voltage in two
- Same as two resistors, but you can vary them



Sensing the Dark: Photocells

- aka. photoresistor, light-dependent resistor
- A *variable* resistor
- Brighter light == lower resistance
- Photocells you have range approx. 0-10k

schematic symbol



Photocell Circuit





Photocell Arduino Sketch

Can use as before, sketch "analog_read_led"

Change to 0

```
int potPin = 2;    // select the input pin for the potentiometer
int ledPin = 13;   // select the pin for the LED
int val = 0;       // variable to store the value coming from the sensor

void setup() {
  pinMode(ledPin, OUTPUT);  // declare the ledPin as an OUTPUT
}

void loop() {
  val = analogRead(potPin);    // read the value from the sensor
  digitalWrite(ledPin, HIGH);  // turn the ledPin on
  delay(val);                  // stop the program for some time
  digitalWrite(ledPin, LOW);   // turn the ledPin off
  delay(val);                  // stop the program for some time
}
```

Wave your hand over it = blink faster
Point it towards the light = blink slower

Moving on…

- Connect a photocell instead of a pot to you fading circuit
  - Do you get the same range of fade as with the pot?
  - Why or why not?



Resistive sensors

thermistor
(temperature)

circuit is the same
for all these



photocell
(light)



force sensors
(pressure)

flex sensor
(bend, deflection)

also air pressure
and others

LED Brightness Functions

Then turn those numbers into an array

```
// the table containing the "curve" the brightness should take
byte bright_table[] = { 30, 30, 30, 40, 50, 60, 70, 80, 90,100,
                        110,120,130,140,150,160,170,180,190,200,
                        210,220,230,240,250,250,240,230,220,210,
                        200,190,180,170,160,150,140,130,120,110,
                        100, 90, 80, 70, 60, 50, 40, 30, 30, 30  };
int max_count = 50;   // number of entries in the bright_table
```

Use any pattern of numbers you like
but they must range between 0-255

```
  0 = full off
127 = half on
255 = full on
```

LED Brightness Functions

Once you have your table...

```
// the table containing the "curve" the brightness should take
byte bright_table[] = { 30, 30, 30, 40, 50, 60, 70, 80, 90,100,
                        110,120,130,140,150,160,170,180,190,200,
                        210,220,230,240,250,250,240,230,220,210,
                        200,190,180,170,160,150,140,130,120,110,
                        100, 90, 80, 70, 60, 50, 40, 30, 30, 30  };
int max_count = 50;   // number of entries in the bright_table
```

...the rest is just programming

1. Get a `bright_table` value
2. Send it out with `analogWrite()`
3. Advance counter into `bright_table`
4. Wait a bit
5. Repeat

Glowing Eyes Sketch

"led_glow"

```
int potPin = 0;
int ledPin = 10;

// the table containing the "curve" the brightness should take
byte bright_table[] = { 30, 30, 30, 40, 50, 60, 70, 80, 90,100,
                        110,120,130,140,150,160,170,180,190,200,
                        210,220,230,240,250,250,240,230,220,210,
                        200,190,180,170,160,150,140,130,120,110,
                        100, 90, 80, 70, 60, 50, 40, 30, 30, 30  };
int max_count = 50;   // number of entries in the bright_table
int count = 0;        // position within the bright_table
int val = 0;          // variable for reading pin status

void setup() {
  pinMode(ledPin, OUTPUT);      // sets the digital pin as output
}

void loop() {
  analogWrite(ledPin, bright_table[count]);   // sets the LED brightness
  count++;                      // moves counter to next position in table
  if( count > max_count )
    count = 0;                  // if at end of table, back to start

  val = analogRead(potPin);
  val = val/4;    // scale it down so it's quicker
  delay(val);
}
```



Communicating with Others

- Arduino can use same USB cable for programming and to talk with computers
- Talking to other devices uses the "`Serial`" commands
  - `Serial.begin()` – prepare to use serial
  - `Serial.print()` – send data to computer
  - `Serial.read()` – read data from computer

Serial Communication

- Serial.begin(baud-rate);
  - baud-rate is 300, 1200, 2400, 4800, 9600, 14400,19200, 28800, 57600, or 115200
  - Sets serial bit rate
- Serial.print(arg);
  - sends arg to the serial output – can be number or string
  - Serial.print(arg,format); // formats the arg
    - format can be BYTE, BIN, OCT, DEC, HEX
- Serial.println(arg);
  - Same, but also prints a newline to the output

Serial Communication

- Serial.available();
  - returns an int that tells you how many bytes remain in the input buffer
- Serial.read();
  - returns the next byte waiting in the input buffer
- Serial.flush();
  - clear the input buffer of any remaining bytes

Serial Example

```
int incomingByte = 0; // for incoming serial data
void setup() {
      Serial.begin(9600); // opens serial port, sets data rate to 9600 bps
}

void loop() {        // send data only when you receive data:
      if (Serial.available() > 0) {   // read the incoming byte:
            incomingByte = Serial.read();

      // say what you got:
      Serial.print("I received: ");
      Serial.println(incomingByte, DEC);
      }
}
```

Arduino Says "Hi"

"SerialHelloWorld"

Sends "Hello world!" to your computer

Click on "Serial Monitor" button to see output

Watch TX LED compared to pin 13 LED



Telling Arduino What To Do

"SerialReadBasic"

You type "H", LED blinks

In "Serial Monitor", type "H", press Send

`Serial.available()` tells you if data present to read

Arduino Communications

is just serial communications

- Psst, Arduino doesn't really do USB
- It really is "serial", like old RS-232 serial
- All microcontrollers can do serial
- Not many can do USB
- Serial is easy, USB is hard

serial terminal from the olde days

Serial Communications

- "Serial" because data is broken down into bits, each sent one after the other down a single wire.
- The single ASCII character 'B' is sent as:

```
'B' =  0 1 0 0 0 0 1 0
    =  L H L L L L H L
    =
```

- Toggle a pin to send data, just like blinking an LED
- You could implement sending serial data with `digitalWrite()` and `delay()`
- A single data wire needed to send data. One other to receive.

Arduino & USB-to-serial

Arduino board is really two circuits



Arduino Mini

Arduino Mini separates the two circuits



Arduino Mini USB adapter



Arduino Mini

Arduino to Computer



USB is totally optional for Arduino
But it makes things easier

Arduino & USB

- Since Arduino is all about serial
- And not USB,
- Interfacing to things like USB flash drives, USB hard disks, USB webcams, etc. is *not* possible

Controlling the Computer

- Can send sensor data from Arduino to computer with `Serial.print()`
- There are many different variations to suite your needs:

```
int val = 123;
Serial.print(val);      // sends 3 ASCII chars "123"
Serial.print(val,DEC);  // same as above
Serial.print(val,HEX);  // sends 2 ASCII chars "7B"
Serial.print(val,BIN);  // sends 8 ASCII chars "01111011"
Serial.print(val,BYTE); // sends 1 byte, the verbatim value
```

Controlling the Computer

You write one program on Arduino, one on the computer

In Arduino: read sensor, send data as byte

```
void loop() {
  val = analogRead(analogInput);   // read the value on analog input
  Serial.print(val/4,BYTE);        // print a byte value out
  delay(50);                       // wait a bit to not overload the port
}
```

In Processing: read the byte, do something with it

```
import processing.serial.*;

Serial myPort;  // The serial port

void setup() {
  String portname = "/dev/tty.usbserial-A3000XV0";
  myPort = new Serial(this, myPort, 9600);
}

void draw() {
  while (myPort.available() > 0) {
    int inByte = myPort.read();
    println(inByte);
  }
}
```

Controlling the Computer

- Receiving program on the computer can be in any language that knows about serial ports
  - C/C++, Perl, PHP, Java, Max/MSP, Python, Visual Basic, etc.
- Pick your favorite one, write some code for Arduino to control

Controlling Arduino, Again

"SerialReadBlink"

Type a number 1-9 and LED blinks that many times

Converts typed ASCII value into usable number

Most control issues are data conversion issues



Ctrl	Dec	Hex	Char	Code	Dec	Hex	Char	Dec	Hex	Char	Dec	Hex	Char
^@	0	00		NUL	32	20		64	40	@	96	60	`
^A	1	01		SOH	33	21	!	65	41	A	97	61	a
^B	2	02		STX	34	22	"	66	42	B	98	62	b
^C	3	03		ETX	35	23	#	67	43	C	99	63	c
^D	4	04		EOT	36	24	$	68	44	D	100	64	d
^E	5	05		ENQ	37	25	%	69	45	E	101	65	e
^F	6	06		ACK	38	26	&	70	46	F	102	66	f
^G	7	07		BEL	39	27	'	71	47	G	103	67	g
^H	8	08		BS	40	28	(	72	48	H	104	68	h
^I	9	09		HT	41	29	)	73	49	I	105	69	i
^J	10	0A		LF	42	2A	*	74	4A	J	106	6A	j
^K	11	0B		VT	43	2B	+	75	4B	K	107	6B	k
^L	12	0C		FF	44	2C	,	76	4C	L	108	6C	l
^M	13	0D		CR	45	2D	-	77	4D	M	109	6D	m
^N	14	0E		SO	46	2E	.	78	4E	N	110	6E	n
^O	15	0F		SI	47	2F	/	79	4F	O	111	6F	o
^P	16	10		DLE	48	30	0	80	50	P	112	70	p
^Q	17	11		DC1	49	31	1	81	51	Q	113	71	q
^R	18	12		DC2	50	32	2	82	52	R	114	72	r
^S	19	13		DC3	51	33	3	83	53	S	115	73	s
^T	20	14		DC4	52	34	4	84	54	T	116	74	t
^U	21	15		NAK	53	35	5	85	55	U	117	75	u
^V	22	16		SYN	54	36	6	86	56	V	118	76	v
^W	23	17		ETB	55	37	7	87	57	W	119	77	w
^X	24	18		CAN	56	38	8	88	58	X	120	78	x
^Y	25	19		EM	57	39	9	89	59	Y	121	79	y
^Z	26	1A		SUB	58	3A	:	90	5A	Z	122	7A	z
^[	27	1B		ESC	59	3B	;	91	5B	[	123	7B	{
^\	28	1C		FS	60	3C	<	92	5C	\	124	7C	|
^]	29	1D		GS	61	3D	=	93	5D	]	125	7D	}
^^	30	1E	▲	RS	62	3E	>	94	5E	^	126	7E	~
^-	31	1F	▼	US	63	3F	?	95	5F	_	127	7F	⌂*

* ASCII code 127 has the code DEL. Under MS-DOS, this code has the same effect as ASCII 8 (BS). The DEL code can be generated by the CTRL + BKSP key.

ASCII codes

Standard byte codes for characters

Mysterious **val = val – '0';** statement converts the byte that represents the character to a byte of that number

For example, if the character is '3', the ASCII code is 51

The ASCII code for '0' is 48

So, 51 – 48 = 3

This converts the character '3' into the number 3

Reading Serial Strings

- The function "`Serial.available()`" makes reading strings easier
- Can use it to read all available serial data from computer
- The "`readSerialString()`" function at right takes a character string and sticks available serial data into it

```
//read a string from the serial and store it in an array
//you must supply the array variable
void readSerialString (char *strArray) {
  int i = 0;
  if(!Serial.available()) {
    return;
  }
  while (Serial.available()) {
    strArray[i] = Serial.read();
    i++;
  }
  strArray[i] = 0;  // indicate end of read string
}
```

Moving on… Servos

- Servo motors are small DC motors that have a range of motion of 0-180º
  - Internal feedback and gearing to make it work
  - easy three-wire interface
  - position is controlled by PWM signals

PWM

Output voltage is averaged from on *vs.* off time

output_voltage = (on_time / off_time) * max_voltage



PWM

- Used everywhere
  - Lamp dimmers, motor speed control, power supplies, noise making
- Three characteristics of PWM signals
  - Pulse width range (min/max)
  - Pulse period (= 1/pulses per second)
  - Voltage levels (0-5V, for instance)



Servomotors

- Can be positioned from 0-180° (usually)
- Internal feedback circuitry & gearing takes care of the hard stuff
- Easy three-wire PWM 5V interface

Servos are Awesome

- DC motor
- High-torque gearing
- Potentiometer to read position
- Feedback circuitry to read pot and control motor
- All built in, you just feed it a PWM signal



Servos, good for what?

- Roboticists, movie effects people, and puppeteers use them extensively
- Any time you need controlled, repeatable motion
- Can turn rotation into linear movement with clever mechanical levers

Servos

- Come in all sizes
  - from super-tiny
  - to drive-your-car
- But all have the same 3-wire interface
- Servos are spec'd by:

weight: 9g
speed: .12s/60deg @ 6V
torque: 22oz/1.5kg @ 6V
voltage: 4.6~6V
size: 21x11x28 mm

9g

157g

Our servos are: weight: 9g,
speed 0.12s/60deg at 4.8v,
torque (@4.8v) 17.5oz/in (1kg/cm)
voltage range: 3.0 – 7.2v

Servo Mounts & Linkages

Lots of ways to mount a servo



And turn its rotational motion into other types of motion



Servo Control



- PWM freq is 50 Hz (i.e. every 20 millisecs)
- Pulse width ranges from 1 to 2 millisecs
  - 1 millisec = full anti-clockwise position
  - 2 millisec = full clockwise position

Servo Movement

0 degrees

1000 microsecs

90 degrees



1500 microsecs

180 degrees



2000 microsecs

In practice, pulse range can range from 500 to 2500 microsecs

Servo and Arduino

First, add some jumper wires to the servo connector



Servo Example Program

```
#include <Servo.h>          // include the built-in servo library
Servo myservo;              // create a servo object to control the servo
int pos = 0;                // variable to store the servo position

void setup() {
    myservo.attach(9);      // attach servo control to pin 9
}

void loop() {
    for(pos = 0; pos < 180; pos++) {    // go from 0 to 180 degrees
        myservo.write(pos);             // move the servo
    }   delay(15);l                     // give it time to get there
    for (pos = 180; pos>=1; pos--) {    // wave backwards
        myservo.write(pos);
        delay(15);
    }
}
```

Servo Functions

- Servo is a class
  - Servo myservo; // creates an instance of that class
- myservo.attach(pin);
  - attach to an output pin (doesn't need to be PWM pin!)
  - Servo library can control up to 12 servos on our boards
  - but a side effect is that it disables the PWM on pins 9 and 10
- myservo.write(pos);
  - moves servo – pos ranges from 0-180
- myservo.read();
  - returns the current position of the servo (0-180)

Moving on…

- Write a program to control the position of the servo from a pot, or from a photocell
  - remember pot analogRead(); values are from 0-1023
  - measure the range of values coming out of the photocell first?
  - use Serial.print(val); for example
  - use map(…); to map values to 0-180

Side Note - Power

- Servos can consume a bit of power
  - We need to make sure that we don't draw so much power out of the Arduino that it fizzles
  - If you drive more than a couple servos, you probably should put the servo power pins on a separate power supply from the Arduino
  - Use a wall-wart 5v DC supply, for example

Robo Cat Toy Idea



Summary – Whew!

- LEDs – use current limiting resistors (remember color code!)
  - drive from digitalWrite(pin,val); for on/off
  - drive from analogWrite(pin,val); for PWM dimming (values from 0-255)
- buttons – current limiting resistors again
  - active-high or active low (pullup or pulldown)
  - read with digitalRead(pin);
- potentiometers (pots)– voltage dividers with a knob
  - use with analogRead(pin); for values from 0-1023

Summary – Whew!

- photocells – variable resistors
  - use with current-limiting resistors (to make voltage divider)
- Serial communications – read a byte, or write a value
  - communicate to the Arduino enviroment, or your own program
- Servos – use Servo library to control motion
  - might need external power supply
  - range of motion 0-180º
- Also `setup( )` and `loop( )` functions, and various C programming ideas

More Later…

- DC Motors
  - use transistors as switches for larger current loads
- Stepper motors
  - Sort of like servos, but with continuous range of motion
  - Can also be more powerful
- I2C serial bus
  - Various LED driver chips
  - other serially-controlled devices
- Piezo buzzers
  - make some noise!
  - But you can also use them as input devices to sense movement
- IR motion sensors
  - simple motion and also distance sensors
- Accelerometers
  - Wii nunchucks, for example
- Others?

Assignment #2

- Form teams
  - three teams – each with one FA3800 student on it
- Check out some supplies
  - Arduino, pots, LEDs, servos, resistors, etc.
- Do something cool!
  - Rather fuzzy specification, but be creative
  - Due next week? Two weeks?